Symbolic SPICE

Symbolic SPICE™

Circuit Analyzer and Approximator

Application Note

AN-005:

An Op-Amp Wien Bridge Oscillator

by Gregory M. Wierzba

Rev 072010

A) Introduction

The schematic shown in Fig. 1 is that of an op-amp Wien bridge oscillator [1]. Oscillators consist of an amplifier and passive network. In this case the passive network contains two capacitors which can provide a phase shift approaching ±90°. At one frequency the phase shift of the passive network is exactly 0° and this is the frequency of oscillation. For the circuit to sustain the oscillation at node 1, the noninverting amplifier must boost the signal by the attenuation created by the passive network as well as to provide the necessary 0° of phase shift to return the signal to node 1 unchanged. So there are two conditions that need to be found to characterize this oscillator. These are the frequency of oscillation and the gain needed by the amplifier for oscillation.

A Symbolic SPICE™ input file, **`wien.cir`**, is given in Table 1. This file uses Symbolic SPICE's definition [2] of an ideal op-amp which begins with the letters `XOA`.



Figure 1. Wien bridge oscillator

Table 1. Symbolic SPICE input file

```
Wien Bridge Oscillator
R 1 2
C 2 3
R 3 0
C 3 0
XOA 3 4 1
R 4 0
R4 4 1
.END
```


B) Running Symbolic SPICE

Running [2] the input file shown in Table 1, the following are the prompts. The user responses are shown in bold:

```
Symbolic SPICE - Circuit Analyzer and Approximator
Demo Version 3.1
(C) Copyright 2010 by Willow Electronics, Inc.
```

```
INPUT FILE NAME [.cir] : wien

OUTPUT FILENAME [wien.det] : (hit enter)

Determinant string sorted according to orders of some variable? (y/n) : n

Numerical evaluation of the results? (y/n) : n

Discard terms if their magnitude falls below a threshold? (y/n) : n

Check and solve for second order filter functions? (y/n) : y

FILTER FUNCTION FILE NAME [wien.fun] : (hit enter)

Solve for a variable or expression? (y/n) : y

Available Unknowns:
V1 V2 V3

*Ignore nodes 5 and higher if present. They are used for internal numbering.

Valid Operators: +, -, *, /, ( ), { }, [ ]

Equation: v1

Solve for another variable or expression? (y/n) : n
```

C) Symbolic SPICE Determinant and Filter Function Output

The output file **wien.det** listed in Table 2 is the matrix written by Symbolic SPICE and the transfer function requested.

Table 2. Symbolic SPICE Output File **wien.det**

```
Wien Bridge Oscillator

[0  ] [-G                      sC+G                     -sC                    ][V1 ]
[0  ]=[0                       -sC                      sC+sC+G                ][V2 ]
[0  ] [-G4                     0                        G4+G                   ][V3 ]

*Ignore nodes 5 and higher if present. They are used for internal numbering.

Numerator of: v1

TERMS SORTED ACCORDING TO POWERS OF s

s**0 terms:

 + 0

************************************************
```



```
Denominator of: v1

TERMS SORTED ACCORDING TO POWERS OF s

s**2 terms:

 - sC*sC*G4

s**1 terms:

 -  2*sC*G4*G + sC*G*G

s**0 terms:

 - G4*G*G

***********************************************
```

In [3], we solved for conditions of oscillation from the characteristic equation. Because the Wien bridge oscillator is a second order system we could use the filter solving option in Symbolic SPICE to also find the conditions of oscillation. The output file **`wien.fun`** is listed in Table 3.

Table 3. Symbolic SPICE Output File **`wien.fun`**

```
Wien Bridge Oscillator

SECOND ORDER FILTER PARAMETERS:

Qo is:

( + G4)
------------
( - 2*G4 + G)*1

Wo**2 is:

( + G*G)
----------
( + C*C)

***************************************

There exists no standard form second order filter function for : v1

***************************************
```

For a second order system we can write the characteristic equation in the following form:

$$s^2 + s\frac{\omega_o}{Q_o} + \omega_o^2 \qquad \textbf{(1)}$$

where Symbolic SPICE will solve for Q_o and ω_o. For an oscillator, the gain for a noise component at the frequency of oscillation must be very large to produce a sinewave output. One way to think about this is that the Q of the circuit is approaching infinity at this frequency. Thus from Table 3, we can solve for the conditions of oscillation.

$$\omega \triangleq \omega_o = \sqrt{\frac{G^2}{C^2}} = \frac{1}{RC}$$

To make Q_o in Table 3 approach infinity

$$-2G_4 + G = 0$$

$$2G_4 = G$$

$$R_4 = 2R$$

Thus the gain of the noninverting amplifier and the frequency of oscillation are

$$1 + \frac{R_4}{R} = 3 \quad \text{and} \quad \omega_o = \frac{1}{RC} \tag{2}$$

D) Wien Bridge Oscillator Design

Suppose we design an oscillator for testing audio equipment. This type of testing is usually done at 1k Hz.

Let C be a standard capacitance value of 0.01μ F then from Eqn. 2, we have

$$R = \frac{1}{0.01\mu \; 2\pi \; 1000} = 15.92\text{k } \Omega$$

The nearest standard value of resistor is 16k Ω. Also from Eqn. 2, we have that

$$R_4 = 2 \bullet 16\text{k} = 32\text{k } \Omega$$

We will need a pot to adjust R_4 to the exact value needed for oscillation. Simulating this in PSpice has some problems in that a transient response rich in harmonics is needed to initiate the oscillation. One way to "kick start" the oscillator is to turn on the power with a switch. We will also need an op-amp model that includes the non-ideal effect of power supply clipping, otherwise our oscillator will produce very large voltages. A SPICE macromodel [4] will satisfy this. An LF411 low distortion op-amp was selected for the noninverting amplifier. The input file is listed in Table 3.

Table 3. PSpice input file **`audio_wien.cir`**

```
Wien Bridge Oscillator
R1 1 2 16K
C1 2 3 0.01U
R2 3 0 16K
C2 3 0 0.01U
R3 4 0 16K
R4 4 1 32K
XOA 3 4 5 6 1 LF411
VCC 5 0 PULSE (0 15 1N)
VEE 6 0 -15
* connections:   non-inverting input
*                |  inverting input
*                |  |  positive power supply
*                |  |  |  negative power supply
*                |  |  |  |  output
*                |  |  |  |  |
.subckt LF411    1  2  3  4  5
*
  c1   11 12 4.196E-12
  c2    6  7 10.00E-12
  css  10 99 1.333E-12
  dc    5 53 dx
  de   54  5 dx
  dlp  90 91 dx
  dln  92 90 dx
  dp    4  3 dx
  egnd 99  0 poly(2) (3,0) (4,0) 0 .5 .5
  fb    7 99 poly(5) vb vc ve vlp vln 0 31.83E6 -30E6 30E6 30E6 -30E6
  ga    6  0 11 12 251.4E-6
  gcm   0  6 10 99 2.514E-9
  iss  10  4 dc 170.0E-6
  hlim 90  0 vlim 1K
  j1   11  2 10 jx
  j2   12  1 10 jx
  r2    6  9 100.0E3
  rd1   3 11 3.978E3
  rd2   3 12 3.978E3
  ro1   8  5 50
  ro2   7 99 25
  rp    3  4 15.00E3
  rss  10 99 1.176E6
  vb    9  0 dc 0
  vc    3 53 dc 1.500
  ve   54  4 dc 1.500
  vlim  7  8 dc 0
  vlp  91  0 dc 25
  vln   0 92 dc 25
.model dx D(Is=800.0E-18 Rs=1m)
.model jx NJF(Is=12.50E-12 Beta=743.3E-6 Vto=-1)
.ends
.PROBE
.TRAN 2.5U 5M 0 2.5U
.END
```

The schematic is shown in Fig. 2 and the output is shown in Fig. 3.



Figure 2. Wien bridge audio oscillator with op-amp macromodel



Figure 3. Wien bridge audio oscillator output

The period in steady-state is 1.0051m sec which is a frequency of 994.9 Hz. Using the values of 16k Ω and 0.01µ F in Eqn. 2, f_o = 994.7 Hz. This is a difference of 0.02 %.

The actual amplitude will vary from that above and depend on the actual parameters of the op-amp.

E) Conclusion

Symbolic SPICE can be used to analyze ideal op-amp circuits to find the conditions of oscillation in a Wien bridge oscillator using second order filter parameters. Symbolic SPICE allows identical elements to have the same SPICE name and symbolic representation. This simplifies the formulas.

F) References

[1] G. M. Wierzba, *ECE 303: Electronics Laboratory Lab Manual*, Lab VII. This e-book is available at http://stores.lulu.com/willowepublishing

[2] G. M. Wierzba, *Symbolic SPICE User Manual*. This e-book is available at http://stores.lulu.com/willowepublishing

[3] G. M. Wierzba, *AN-004: An Op-amp Phase Shift Oscillator*. This application note is available at www.willowelectronics.com

[4] G. M. Wierzba, *ECE 402: Application of Analog ICs Class Notes,* Ch.5. This e-book is available at http://stores.lulu.com/willowepublishing